

# Some Like it Tepid!

TI-84 Plus CE with TI-Innovator™ Hub and TI-SensorLink™ Adapter

**Project Overview:** In this activity, students are challenged to design a system that informs the user if their coffee is too hot to drink. The system should notify the user of three states: Too hot to drink; too cold to drink; and just right to drink. Students may choose to notify the user with a particular color of the TI-Innovator Hub's onboard LED and/or a particular sound played on the built-in speaker. After designing and building the system, use the hot water and then ice water to test the system. Use the results of your tests to adjust the code to ensure the system works as expected.



## Parts List for project:

- TI-Innovator™ Hub (requires Sketch 1.3 or above)
- TI 84 Plus CE
- TI-SensorLink™ Adapter
- Vernier BTA SS Temperature Probe
- Coffee cup or styrofoam cup
- Hot water
- Ice cubes in water
- Instant coffee (optional)

## Example TI-BASIC Code for TI 84 Plus CE:

1. Code snippet to set the RGB color LED to yellow on the Hub:

```
Send("SET COLOR 255 128 0")
```

2. Code snippet to set speaker to 256 Hz on the Hub for 10 seconds:

```
Send("SET SOUND 256 TIME 10")
```

3. Code snippet to display a message on the calculator:

```
ClrHome
Disp "JUST RIGHT"
```

4. Code snippet to connect and read the temperature sensor:

```
Send("CONNECT VERNIER 1 TO IN1 AS TEMPERATURE")
Send("READ VERNIER 1")
Get(T)
Disp "TEMPERATURE=", round(T,1)
```

Note: "VERNIER" must be typed into the program using the keypad.

5. Code snippet to read and display the temperature sensor continuously every second until the [enter] key is pressed.

```
0→K
While K≠45
getKey→K
Send("READ VERNIER 1")
Get(T)
Disp "TEMPERATURE=", round(T,1)
Wait 1
End
```

6. Code snippet to make a decision tree using the If-Then-End statement:

```
If T>65
Then
Disp "TOO HOT"
Send("SET COLOR 255 0 0")
Send("SET SOUND 1024 TIME 10")
End
If T≥50 and T≤65
Then
Disp "JUST RIGHT"
Send("SET COLOR 0 255 0")
Send("SET SOUND 512 TIME 10")
End
If T<50
Then
Disp "TOO COLD",
Send("SET COLOR 0 0 255")
Send("SET SOUND 256 TIME 10")
End
```

## 7. Complete Program:

```
Send("CONNECT VERNIER 1 TO IN1 AS TEMPERATURE
ClrHome
Disp "HOLD [clear] TO QUIT"
Wait 2
0→K
While K≠45
Send("READ VERNIER 1")
Get(T)
ClrHome
Disp "TEMPERATURE=",round(T,1)
If T>65
Then
Disp "TOO HOT"
Send("SET COLOR 255 0 0")
Send("SET SOUND 1024 TIME 10")
End
If T≥50 and T≤65
Then
Disp "JUST RIGHT"
Send("SET COLOR 0 255 0")
Send("SET SOUND 512 TIME 10")
End
If T<50
Then
Disp "TOO COLD",
Send("SET COLOR 0 0 255")
Send("SET SOUND 256 TIME 10")
End
getKey→K
Wait 1
End
```